# $Id: PKGBUILD 88656 2013-04-21 08:48:16Z jelle $
# Contributor: Paulo Matias <matiasΘarchlinux-br·org>
# Maintainer: Jelle van der Waa <jelle@vdwaa.nl>

pkgname=pdf2djvu
pkgver=0.7.16
pkgrel=4
pkgdesc="Creates DjVu files from PDF files"
arch=('i686' 'x86_64')
url="http://pdf2djvu.googlecode.com"
license=('GPL')
depends=('poppler' 'djvulibre' 'libxslt' 'gcc-libs' 'graphicsmagick')
makedepends=('pstreams' 'python2-nose' 'ttf-liberation')
source=("http://pdf2djvu.googlecode.com/files/${pkgname}_${pkgver}.tar.gz")

build() {
    cd ${srcdir}/${pkgname}-${pkgver}
    ./configure --prefix=/usr
    make
}

package() {
    cd ${srcdir}/${pkgname}-${pkgver}
    make install DESTDIR=${pkgdir}
    install -Dm644 doc/${pkgname}.1 ${pkgdir}/usr/share/man/man1/${pkgname}.1
}
check() {

    cd ${srcdir}/${pkgname}-${pkgver}
    sed -i 's/nosetests/nosetests2/' tests/Makefile
    make test
}
md5sums=('6ba1e576212e129111518a8c4afd1092')
